

## Module 9 Plaatjes in twee dimensies

---

<b>Onderwerp</b>	Onderzoek van (grafieken van) functies en (vlakke) krommen.
<b>Voorkennis</b>	VWO-stof over functieonderzoek.
<b>Expressies</b>	<code>plot</code> , <code>display</code> , <code>implicitplot</code> , <code>polarplot</code> , <code>listplot</code> , <code>pointplot</code> , <code>textplot</code> , <code>cat</code> , <code>convert</code> , <code>string</code> , <code>animate</code> , <code>animatecurve</code>
<b>Bibliotheken</b>	<code>plots</code> , <code>plottools</code>
<b>Bestanden</b>	<code>meting1.dat</code>
<b>Zie ook</b>	Module 8.

---

### 9.1 Het tekenen van grafieken

**Eén grafiek.** We kunnen een grafiek tekenen van een functie  $f$  met het commando `plot`. Als we van  $f(x)$  het deel van de grafiek willen zien waarbij  $x$  ligt in het interval  $[a, b]$ , gebruiken we

```
plot(f(x), x=a..b);
```

of `plot(f, a..b);`. Net als bij `diff`, `int` en `limit` verwacht `plot` als eerste argument een uitdrukking die niet noodzakelijk van de vorm  $f(x)$  hoeft te zijn. De grafiek van de functie

$$f(x) = x^2 - \frac{1}{2}, \quad x \in [-1, 1]$$

kunnen we op de volgende manieren verkrijgen:

- `plot( x^2 - 1/2, x=-1..1 );`
- `f := x->x^2-1/2: plot( f, -1..1 );`
- `plot( x->x^2-1/2, -1..1 );`

Maple zorgt ervoor dat de verticale as lang genoeg is voor alle functiewaarden op het interval  $[a, b]$  dat voor de horizontale as is opgegeven. Dat is niet altijd gewenst, bijvoorbeeld wanneer  $f$  in het interval  $[a, b]$  een verticale asymptoot heeft. Daarom kunnen we ook op de *verticale* as het interval aangeven dat we willen bekijken. Dit gaat dan met

```
plot(f(x), x=a..b, y=c..d);
```

of met `plot(f, a..b, c..d);`.

Als u met de cursor *op* het plaatje gaat staan, dan krijgt u met de rechter muisknop een menu waarmee u nog verschillende dingen aan het plaatje kunt veranderen. U kunt hiermee ook het plaatje als bestand opslaan, zodat u het in een ander document kunt opnemen.

Meer mogelijkheden om een grafiek te verfraaien vindt u onder **T**ools → **A**ssistants → **P**lot Builder. Het is interessant om als u alle dialoogvensters in deze *Plot Builder* hebt afgewerkt (inclusief Options) niet af te sluiten met **P**lot, maar met **C**ommand. U krijgt dan als (blauwe) *uitvoer* een Maplecommando waarmee de tekening wordt gemaakt. Hierin komen allerlei *opties* voor, waarvan de meeste in §9.4 worden uitgelegd. Als u deze tekening ook te zien wilt krijgen, moet u deze output wel eerst even kopiëren naar een (rode) inputregel. Als u afsluit met **P**lot, dan krijgt u direct het plaatje, zonder dat het bijbehorende commando zichtbaar wordt.<sup>23</sup>

**Meer grafieken in één plaatje.** Om twee of meer *grafieken van functies* in één plaatje te laten tekenen, kunnen we gewoon een *lijst* (eventueel een verzameling) van te tekenen grafieken opgeven. Met

```
plot( [x^2,x^3,x^4], x=-1..1 );
```

worden de verwachte drie grafieken in één plaatje getekend. Maple zoekt zelf drie verschillende kleuren uit. Als u daar niet tevreden mee bent:

```
plot( [x^2,x^3,x^4], x=-1..1, color=[red,blue,black] );
```

om de grafiek van  $x^2$  rood te maken, die van  $x^3$  blauw, en die van  $x^4$  zwart. Als u `color=red` opgeeft, worden ze alle drie rood.

## 9.2 Het tekenen van geparametriseerde en impliciet gegeven krommen

**Parameterkrommen.** Ook van een kromme, gegeven door een *parametervoorstelling*, kan gemakkelijk een plot gemaakt worden. Als zo'n kromme gegeven is door  $(x(t), y(t))$ , dan wordt een plot verkregen (voor  $t$  tussen  $t_0$  en  $t_1$ ) met de opdracht

```
plot( [x(t), y(t), t=t0..t1] );
```

Let op de vierkante haken; merk op dat het interval voor de parameter  $t$  óók tussen de vierkante haken geplaatst moet worden. Als we dat niet zouden doen, dan zou Maple de bovenstaande opdracht interpreteren als: “teken twee grafieken, namelijk die van  $x(t)$  en  $y(t)$ , met de  $t$ -waarden langs de horizontale en de  $x$ - en  $y$ -waarden langs de verticale as.”

<sup>23</sup>Het nadeel hiervan is echter dat u bijvoorbeeld na **!!!** (execute worksheet) uw plaatje kwijt bent.

Ook nu kunnen intervallen die we willen bekijken voor de  $x$ - en de  $y$ -coördinaten worden opgegeven:

```
plot( [x(t), y(t), t=t0..t1], x=a..b, y=c..d );
```

Als we dat niet doen, dan zorgt Maple er zelf voor dat de assen zó lang zijn dat de hele kromme in het plaatje past.

Op precies dezelfde manier als in §9.1 kunnen diverse krommen in één figuur worden getekend.

### Voorbeeldopgave


Teken in één figuur de krommen gegeven door

$$\begin{cases} x(t) = \cos(t) \\ y(t) = \sin(3t) \end{cases} \quad \begin{cases} x(t) = \cos(t) \\ y(t) = \sin(5t) \end{cases}$$

Neem hierbij voor  $x$  en  $y$  het interval  $[-\frac{6}{5}, \frac{6}{5}]$ .

### Voorbeeldsessie

```
> c1 := [cos(t), sin(3*t), t=0..2*Pi];
      c1 := [cos(t), sin(3 t), t = 0..2 pi]
> c2 := [cos(t), sin(5*t), t=0..2*Pi];
      c2 := [cos(t), sin(5 t), t = 0..2 pi]
> plot( [c1,c2], x=-1.2..1.2, y=-1.2..1.2, color=[red,blue] );
```



### Toelichting

De te tekenen krommen zijn  $c1$  en  $c2$  die worden gedefinieerd als

*lijsten* met drie elementen, namelijk de  $x$ -coördinaat, de  $y$ -coördinaat en het interval (de *range*) van de parameter.  $\diamond$

**Andere toepassingen van parameterkrommen.** Een zeer eenvoudige parametervoorstelling, namelijk  $(x, f(x))$ , kan worden gebruikt om de grafiek van  $f : [a, b] \rightarrow \mathbb{R}$  uitsluitend op het interval  $[a, b]$  te tekenen, terwijl de  $x$ -as verder doorloopt. Bijvoorbeeld, de grafiek van  $f(x) = x^2$  wordt op het interval  $[-1, 1]$  getekend in een plaatje waarin de  $x$ -as van  $-2$  tot  $2$  loopt met het commando:

```
plot( [t, t^2, t=-1..1], x=-2..2 );
```

We kunnen dat zelfs combineren met een ‘gewone’ grafiek:

```
plot( [[t, t^2, t=-1..1], x^2], x=-2..2 );
```

De parametervoorstelling kan ook worden gebruikt om  $x$  als functie van  $y$  te tekenen. Met:

```
plot( [y^2, y, y=-1..1], x=0..1 );
```

wordt een ‘liggende’ parabool getekend.

implicitplot

**Impliciet gedefinieerde krommen.** De procedure `implicitplot` dient ervoor om een kromme in het vlak die gegeven is door een *vergelijking* zichtbaar te maken. Deze procedure zit in de bibliotheek `plots`.

plots

Stel dat  $f$  een uitdrukking is in  $x$  en  $y$ , zodanig dat door de vergelijking  $f = 0$  een kromme  $K$  wordt vastgelegd. Dan geeft

```
implicitplot( f=0, x=a..b, y=c..d );
```


dat deel van  $K$  (ten opzichte van een assenstelsel  $Oxy$ ) waarvoor  $x \in [a, b]$  en  $y \in [c, d]$ . We moeten wel eerst de bibliotheek `plots` laden met `with(plots)`:

### Voorbeeldopgave

Teken de kromme vastgelegd door de vergelijking  $x^2 + y^4 = 1$ .

### Voorbeeldsessie

```
> implicitplot(x^2 + y^4 = 1, x=-1..1, y=-1..1);
      implicitplot (x^2 + y^4 = 1, x = -1...1, y = -1...1)
> use plots in
  implicitplot(x^2 + y^4 = 1, x=-1..1, y=-1..1)
end use;
Andere mogelijkheden:
> restart; with(plots):
> implicitplot(x^2 + y^4 = 1, x=-1..1, y=-1..1);
> restart;
> plots:-implicitplot(x^2 + y^4 = 1, x=-1..1, y=-1..1);
```



### Toelichting

(Uiteraard hebben we het plaatje maar één keer afgedrukt).

Na het eerste statement zien we wat er gebeurt als we vergeten te vermelden dat de procedure `implicitplot` in de bibliotheek `plots` zit: Maple doet net alsof het `implicitplot` niet kent, en geeft de invoer ongewijzigd terug.

We geven daarna drie verschillende mogelijkheden om een procedure (in dit geval `implicitplot`) uit een bibliotheek (in dit geval `plots`) te gebruiken.

- Tussen `use plots in` en `end use` kunnen we alle commando's uit de bibliotheek `plots` gebruiken. Nuttig als u hem niet de hele sessie nodig hebt.
- Het commando `with(plots)`, meestal direct na een `restart`, gebruikt men als men de hele sessie over alle commando's uit deze bibliotheek wil beschikken.
- met `plots:-implicitplot` wordt `implicitplot` als het ware éénmalig 'uit de bibliotheek gehaald'.

Meestal zult u een of meer bibliotheken standaard aan het begin van uw werkblad laden met `with( )`-commando's. ◇

Als de tekening er wat 'hoekig' of 'bibberig' uiziet, kunt u dat met de optie `numpoints` misschien enigszins verhelpen. Zie blz. 127.

**Polaire krommen.** Eveneens in de bibliotheek `plots` zit de procedure `polarplot`. Met het commando

`polarplot`

```
polarplot( f(theta), theta=0..2*Pi )
```

wordt een kromme getekend waarbij  $f(\theta)$  de afstand tot de oorsprong is, als functie van de hoek  $\theta$  met de positieve  $x$ -as. In feite is dit gewoon een kortere schrijfwijze voor de geparametriseerde kromme

```
plot( [ f(theta)*cos(theta), f(theta)*sin(theta),
        theta=0..2*Pi ] )
```

### 9.3 Teksten, punten en lijnen

**Lijnen.** Als we een rechte lijn willen tekenen, dan kunnen we dat natuurlijk doen als grafiek van een functie van de vorm  $x \mapsto ax + b$ . Eenvoudiger is echter om het beginpunt  $(x_1, y_1)$  en het eindpunt  $(x_2, y_2)$  op te geven, en Maple de opdracht te geven daar een rechte lijn tussen te tekenen. Zo tekent

```
plot([[1,-2],[1,2]]);
```

een verticale lijn bij  $x = 1$  en  $y$  van  $-2$  tot  $2$ . Let op dat *alle* haakjes in dit commando nodig zijn: we tekenen een lijst van punten, waarbij de coördinaten van elk punt als een lijstje van twee getallen is gegeven. Het is ook mogelijk méér dan twee punten met elkaar te verbinden. Met

```
plot([[0,0],[0.5,sin(0.5)],[1,sin(1)],
      [1.5,sin(1.5)],[2,sin(2)]]);
```

wordt een (nogal grove) benadering van de grafiek van de sinusfunctie op het interval  $[0, 2]$  getekend.

**Lijsten en Arrays.** Na het laden van de `plots`-bibliotheek tekent

`listplot`

```
listplot(L)
```

de grafiek van de lijst  $L$  als een gebroken kromme die de punten  $(1, L[1]), (2, L[2]), \dots, (n, L[n])$  met elkaar verbindt.

Inplaats van een lijst mag  $L$  ook een Array (of een Vector, zie Module 13) zijn. Hiervan geven we een voorbeeld.


#### Voorbeeldsessie

```
> with(plots):
> L := [-0.1,0.1,0.5,0.3,0.6,0.5,0.5]:
> listplot(L);
```

(zie figuur 1)

`pointplot`

**Punten.** De `plots`-bibliotheek bevat de procedure `pointplot` waarmee op dezelfde wijze als boven in het ‘sinus-voorbeeld’ de vijf losse punten worden getekend. Dit commando kan bijvoorbeeld goed worden gebruikt om, gebruikmakend van `display`, zie blz. 122, een serie meetpunten samen met een grafiek die het theoretische verband aangeeft in één figuur te tekenen.




FIGUUR 1. Resultaat van listplot

### Voorbeeldsessie

```
> with(plots):
> f := x -> x^3-x+1/2;
      f := x ↦ x3 - x + 1/2
> functie := plot(f,-1..1):
> meetwaarden := [[-.900, .721], [-.500, .879],
[0.020, .420], [.200, .338], [.550, 0.016], [.950, .447]]:
> meetpunten := pointplot( meetwaarden,
symbol=solidcircle, symbolsize=20 ):
> display( {functie,meetpunten} );
```

(zie figuur 2)



FIGUUR 2. Een pointplot samen met een grafiek

## Toelichting

Zie blz. 122 voor uitleg over het `display`-commando. ◇

Met een lijst werkt `pointplot` niet. Om de elementen van een lijst als losse punten te tekenen, moet men `listplot` gebruiken met de optie `style=point`. Zie verder §9.4.

`textplot`

**Teksten.** De procedure `textplot` in de bibliotheek `plots` biedt de mogelijkheid om teksten in een tekening op te nemen. Het commando luidt:

```
plots:-textplot([1,2,"Grafiek van f"]);
```

Hiermee wordt de tekst "grafiek van f" op het punt (1,2) in een plot geplaatst. Overigens mag het begrip 'tekst' hier ruim worden opgevat: elke zinvolle Maple-expressie kan hier worden ingevuld. Bijvoorbeeld

```
a := 1.25: plots:-textplot([1,2,'a'=a]);
```

plot de tekst `a=1.25` op de plaats  $x = 1$ ,  $y = 2$ .

`align`

Met een `align`-optie (zie de voorbeeldsessie of raadpleeg `?align`) kan deze tekst nog enigszins worden verschoven ten opzichte van het opgegeven punt. Als deze optie niet wordt gebruikt, is het opgegeven punt het midden van de tekst (in het geval dus het cijfer 1).

### Grafieken, teksten, punten en lijnen in één tekening.

`display`

Om dergelijke objecten, die met verschillende plot-commando's worden gemaakt samen in een tekening te zetten is het commando `display` bedoeld. Dat werkt als volgt.

- Lees de bibliotheek `plots` in met `with(plots)`: Dit hoeft per sessie<sup>24</sup> maar één keer te gebeuren.
- Ken het resultaat van elk plot-commando toe aan een variabele. Bijvoorbeeld:

```
grafieken := plot( [f(x),g(x)], x=a..b,
                  color=[red,blue] );
punten := pointplot( puntenlijst );
teksten := textplot( [[x1,y1,"f(x)"],
                    [x2,y2,"g(x)"]] );
```

Denk er aan dat hier een dubbele punt in plaats van een puntkomma gebruikt wordt.

- Maak alles zichtbaar met

```
display( {grafieken, punten, teksten}, opties );
```

<sup>24</sup>Beter: na iedere restart.



Denk er aan dat u de te tekenen objecten tussen accolades zet als u nog verdere teken-opties wilt meegeven.


Welke opties allemaal mogelijk zijn, wordt besproken in §9.4.

We geven een voorbeeld waarin diverse mogelijkheden voor het plaatsen van teksten worden gedemonstreerd.

### Voorbeeldsessie

```
> restart; with(plots):
> p1 := plot( exp(x), x=-3..3, y=-3..3,
color=red, thickness=2 ):
> p2 := plot( ln(x), x=0..3, y=0..3,
color=blue, thickness=2):
> p3 := textplot( [1, exp(1), exp(x)],
align={RIGHT,BELOW}, font=[TIMES,16] ):
> p4 := textplot( [2, ln(2), typeset("inverse: ",ln(x))],
align={LEFT,ABOVE}, font=[TIMES,ITALIC,12] ):
> display( {p1,p2,p3,p4}, scaling=constrained,
labels=["x","f(x)"], labelfont=[HELVETICA,10],
labeldirections=[horizontal,vertical], titlefont=[TIMES,16],
title="Een functie en zijn inverse",
caption=typeset(exp(x)," (rood), ", ln(x)," (blauw).") );
```

(zie figuur 3)



FIGUUR 3. Zie de voorbeeldsessie op blz. 123

### Toelichting

Het plaatje bestaat uit vier elementen: twee grafieken (namelijk die

typeset

van  $e^x$  en die van  $\ln x$ ) en twee teksten (namelijk  $e^x$  en *inverse*:  $\ln(x)$ ). Het eerste argument van `textplot` staat tussen vierkante haakjes en bevat achtereenvolgens de  $x$ -coördinaat, de  $y$ -coördinaat en de tekst die op deze plaats moet worden neergezet (als maple-expressie als het een formule is, tussen *string-quotes* als het een gewone tekst is, en met een `typeset` als tekst en formules worden gecombineerd). De `align`-optie geeft aan hoe de tekst ten opzichte van deze plaats moet worden gepositioneerd.

Ook voor het boven- en onderschrift kan `typeset` worden gebruikt om gewone tekst en formules te combineren.

Merk op dat sommige opties bij de *onderdelen* van het plaatje horen, zoals de lijndikte en de kleur van de grafieken. Deze worden dan bij het betreffende `plot`-commando opgegeven. Andere opties horen bij het *hele* plaatje, zoals de labels bij de assen en de titel van het plaatje. Deze opties kunnen bij het `display`-commando worden opgegeven.◊

plottools

**Andere figuren.** De bibliotheek `plottools` bevat een groot aantal procedures die gebruikt kunnen worden om pijlen, cirkels, rechthoeken enzovoort te tekenen. Wij behandelen dit hier niet; raadpleeg `plottools` voor informatie.

## 9.4 Opties

Het `plot`-commando is een Maple-procedure die wordt aangeroepen met één of meer argumenten. Het eerste argument, namelijk ‘datgene dat geplotted moet worden’<sup>25</sup> is om begrijpelijke reden altijd verplicht. Het kortst mogelijke werkende `plot`-commando is: `plot(1)`, waarmee de grafiek van de constante functie  $x \mapsto 1$  op het interval  $[-10, 10]$  wordt getekend.

**Ranges.** Als het eerste argument een formule is, zoals bijvoorbeeld  $x^2$ , of een lijst van formules tussen vierkante haken, bijvoorbeeld `[x^2, x^3, x^4]`, waarin één variabele voorkomt, dan moeten we het *horizontale* interval (de *range*)  $[a, b]$  als *tweede* argument opgeven in de vorm `x=a..b`.<sup>26</sup> Het eventuele *verticale* interval (niet verplicht) moet als *derde* argument worden gegeven.

<sup>25</sup>Overigens is bijvoorbeeld `[sin(t), cos(t), t=-Pi..Pi]` door de gebruikte vierkante haken ook één ‘object’, namelijk een (geparametriseerde) cirkel.

<sup>26</sup>De geparametriseerde kromme `[x(t), y(t), t=a..b]` vormt hierop een uitzondering; hierbij is het tweede argument niet nodig.

De overige argumenten zijn zogenaamde *opties* en zijn niet verplicht. De volgorde van de opties is willekeurig. Wij zijn bijvoorbeeld de `color`-optie al tegengekomen. Deze opties zijn bedoeld om aan Maple door te geven *hoe* het plaatje moet worden getekend. Als we ze weglaten kiest Maple zélf, bij `plot(1)` bijvoorbeeld dat de grafiek als een rode ononderbroken lijn wordt getekend.

Alle opties zijn van de vorm *optienaam=waarde*. Bijvoorbeeld met `plot(1, color=blue)`; geven we aan dat de horizontale lijn niet rood maar blauw moet worden getekend.

**Discontinuïteiten.** Maple tekent een grafiek door voor een groot aantal  $x$ -waarden de bijbehorende functiewaarden te berekenen, en deze punten door rechte lijnen met elkaar te verbinden. Als de functie niet overal continu is, heeft dat tot gevolg dat een punt net links van een discontinuïteit wordt verbonden met een punt juist rechts ervan. Hierdoor worden in de sprongpunten (bijna) verticale lijnstukken getekend.

`discont`

Als dat ongewenst is, moet de optie `discont=true` worden gegeven. Maple gaat dan, voordat de grafiek wordt getekend, eerst onderzoeken of de te tekenen functie op het opgegeven interval discontinuïteiten of verticale asymptoten heeft en waar deze zich bevinden. Het zorgt er dan voor dat de punten links en rechts daarvan niet met elkaar worden verbonden.

**De assen.** Als we voor de assen geen opties meegeven, dan tekent Maple een ‘normaal’ assenstelsel en zet er zelf een schaalverdeling op. Wanneer als tweede en derde argument de ranges zijn aangegeven als `t=a..b`, ‘ $f(t)$ ’=`p..q`, dan zet Maple de letter  $t$  bij de horizontale as en  $f(t)$  bij de verticale as.

`options`

Met verschillende opties kunnen we (bijna) alles veranderen. We behandelen hier de voornaamste. Raadpleeg `?plot[options]` voor meer mogelijkheden.

`axes`

De optie `axes` met mogelijke waarden `FRAME`, `BOXED` en `NONE` kunnen we de assen op een andere plaats zetten. Aanbevolen als u vermoedt dat (een deel van) de grafiek met een van de assen samenvalt zodat het bij een normaal assenstelsel onzichtbaar is.

`labels`





De namen bij de assen kunnen worden veranderd met de optie `labels`:

```
labels=["naam horizontale as","naam verticale as"]
```

Gebruik stringquotes voor gewone tekst. Met

```
labels=[x, exp(x)]
```

(maple-expressies) krijgt u respectievelijk  $x$  en  $e^x$  langs de horizontale en verticale as. Gebruik `typeset` om tekst en formules te combineren:



```
labels=[typeset("input ",x), typeset("output ",exp(x))]
```

Als u langs de verticale as een tamelijk lange naam wil zetten, dan zou u hem ook *verticaal* kunnen nemen (dus een kwartslag gedraaid).

**labeldirections** Dat gaat met de optie `labeldirections`:

```
labeldirections=[horizontal,vertical]
```

De *schaalverdeling* langs de assen is te beïnvloeden met de optie `tickmarks`. Bijvoorbeeld

**tickmarks**

```
tickmarks=[[ -1,0,2], [-0.5,0.5]]
```

zet deelstreepjes langs de horizontale as bij de punten  $-1, 0, 2$  (dus *niet* bij  $x = 1$ ) en langs de verticale as bij  $-0.5$  en  $0.5$ . We kunnen deze optie ook toepassen als `tickmarks=[4,5]`, maar dat heeft een veel minder voorspelbaar resultaat. Hierdoor worden langs de horizontale as *bij benadering* 4 en langs de verticale as 5 deelstreepjes gezet, maar het kan er ook eentje meer of minder worden.

Met `tickmarks=[[ ], [ ]]` of `tickmarks=[0,0]` wordt er dus helemaal geen schaalverdeling op de assen gezet.

Met de optie `tickmarks` kan men niet alleen opgeven waar de deelstreepjes langs de assen komen te staan, maar ook wat er bij deze deelstreepjes wordt vermeld. Vaak zal het voorkomen dat men veelvouden van  $\pi$  langs de horizontale as wil uitzetten. Dat kan met `spacing`. Door

```
plot( sin, -2*Pi..2*Pi,
      tickmarks=[spacing(Pi/2), spacing(1/2)] );
```

krijg je een sinusfunctie te zien met langs de horizontale as:  $-2\pi, -\frac{3\pi}{2}, -\pi$ , enzovoort. Door `spacing(1/2)` voor de verticale as komt daar  $\frac{1}{2}$  te staan inplaats van  $0.5$ . Als we het in gewone kommagetallen hadden willen hebben, dan hadden we daar `spacing(0.5)` kunnen gebruiken, of het woord `default` als Maple het zelf mag uitzoeken.

Dit waren min of meer standaardmogelijkheden. Men kan ook op onregelmatige afstanden verschillende dingen op de assen aangeven. Dat komt er uiteraard wat ingewikkelder uit te zien. Bijvoorbeeld

```
tickmarks=[default, [0.5="minimum", 3.7="maximum"]]
```

zet twee streepjes langs de verticale as, op  $y = 0.5$  voorzien van het woord "minimum" en op  $y = 3.7$  het woord "maximum". Langs de horizontale as worden de standaard-deelstreepjes gezet.

**axis**

Om méér voor de assen mee te geven, moeten we de assen de assen afzonderlijk beschrijven met de optie `axis`<sup>27</sup>, bijvoorbeeld als we ex-


<sup>27</sup>Enkelvoud! Niet verwarren met `axes`

tra deelstreepjes willen en/of een afwijkende kleur willen gebruiken.

Met

```
plot( cos, -2*Pi..2*Pi, color=blue,
      axis[1]=[tickmarks=[spacing(Pi/2),subticks=4],
              color=red] );
```

wordt een *blauwe* cosinusgrafiek getekend, met op de *rode* horizontale as 4 extra deelstreepjes tussen de veelvouden van  $\frac{\pi}{2}$ . De verticale as zoekt Maple zelf uit, maar die zouden we natuurlijk met `axis[2]` kunnen beïnvloeden. Zie figuur 4.



FIGUUR 4. Illustratie `axis`-optie

`color`

**De grafieken.** Allereerst kunnen we de *kleur* waarin de grafiek wordt getekend beïnvloeden met de `color`-optie. Bij een plot met één grafiek ligt het gebruik voor de hand. Raadpleeg eventueel `?plot,color` voor een lijst van mogelijke kleuren. Als er meer grafieken in één plaatje moeten worden getekend, dan kunnen we dat met verschillende `plot`-opdrachten doen, elk met een eigen kleur, en deze met `display` in één plaatje zetten. Een andere mogelijkheid is:

```
plot( [sin(t),cos(t)], t=-Pi..Pi, color=[red,blue] );
```

waarmee een rode sinusgrafiek en een blauwe cosinusgrafiek wordt getekend. Als u er ook nog bij zet

`legend`

```
legend=["sin(t)","cos(t)"]
```

dan vermeldt Maple dit onder de tekening.

`numpoints`

Als de grafiek ‘hoekiger’ is dan u verwacht, dan kunt u hem gladder krijgen door Maple meer punten te laten berekenen met de optie `numpoints=n`. Hierin is *n* het *minimale* aantal functiewaarden dat

	wordt berekend om de grafiek te tekenen. Bij gewone, met het <code>plot</code> -commando getekende is de standaardwaarde $n = 50$ , maar dat zult u haast nooit hoeven te verhogen omdat altijd extra functiewaarden worden berekend waar de functie erg krom is. Bij <code>implicitplot</code> is dat anders. Daar worden standaard (precies) 25 waarden van $x$ en 25 waarden van $y$ in de vergelijking ingevuld (dus $n = 625$ ). De punten die het best aan de vergelijking voldoen, worden dan met elkaar verbonden. Vooral als de kromme zichzelf ergens snijdt of een knik heeft, kan dit nogal onnauwkeurig worden. In dat geval moet men <code>numpoints</code> ophogen (of een parametrisering zien te vinden).
<code>linestyle</code>	Met <code>linestyle</code> kunnen andere dan doorlopende krommen worden getekend. Mogelijke waarden: <code>DOT</code> , <code>DASH</code> en <code>DASHDOT</code> . Verder is er de
<code>thickness</code>	<code>thickness</code> -optie met waarden 0, 1, 2, 3 om dikkere lijnen te tekenen.
<code>style</code>	De optie <code>style=POINT</code> heeft het effect dat alléén de door Maple getekende punten worden getekend (dus zoals bij <code>pointplot</code> ). Dat zijn juist de punten die door rechte lijnstukjes met elkaar worden verbonden als deze optie niet wordt gebruikt.
<code>symbol</code>	Bij <code>pointplot</code> kan men de optie <code>style=LINE</code> gebruiken om de getekende punten juist wél met elkaar te verbinden (zodat hetzelfde effect wordt bereikt als wanneer <code>plot</code> in plaats van <code>pointplot</code> zou zijn gebruikt). Interessanter bij <code>pointplot</code> is de <code>symbol</code> -optie, met als mogelijke waarden: <code>asterisk</code> , <code>BOX</code> , <code>solidbox</code> , <code>CROSS</code> , <code>diagonalcross</code> , <code>CIRCLE</code> , <code>solidcircle</code> , <code>DIAMOND</code> , <code>soliddiamond</code> en <code>POINT</code> . <sup>28</sup> Te gebruiken als bijvoorbeeld verschillende meetseries in één figuur moeten worden weergegeven. De grootte van de gebruikte symbolen kan worden veranderd met de optie <code>symbolsize=n</code> , met $n = 10$ de normale grootte.
<code>symbolsize</code>	
<code>title</code>	<b>Titel.</b> Boven een plaatje kan een <i>titel</i> worden getoond met de optie <pre>title="Grafiek van f(x)"</pre> De tekst in de <code>title</code> -optie <i>moet</i> tussen string-quotes (") staan. Als we er de waarde van een of meer variabelen of formules in willen opnemen, dan moeten we daar een <code>typeset</code> voor gebruiken, zie blz. 124.
<code>caption</code>	De optie <code>caption</code> werkt precies hetzelfde als <code>title</code> , maar zet de tekst <i>onder</i> de grafiek.
	Tenslotte nog een tweetal bijzondere soorten plaatjes.
<code>filled</code>	<b>'Gevulde' gebieden.</b> Als in een <code>plot</code> -commando de optie <code>filled=true</code> wordt gegeven, dan wordt het gebied tussen de grafiek en de horizontale as opgevuld met een eventueel zelf te kiezen kleur.

---


<sup>28</sup>Hoofdletters of kleine letters, maakt niet uit.

### Voorbeeldopgave

Teken de grafiek van de functie  $f(x) = -x^2 + 4x - 2$  en kleur het gebied tussen de  $x$ -as en het positieve deel van de grafiek in.

### Voorbeeldsessie

```
> p := -x^2 + 4*x - 2;
      p := -2 - x^2 + 4x
> solve(p): X := sort([%]);
      X := [2 - sqrt(2), 2 + sqrt(2)]
> gebied := plot( p, x=X[1]..X[2], color=gray, filled=true );
> grafiek := plot(p, x=0..4, color=black, thickness=3 );
> plots[display]({grafiek,gebied});
```



### Toelichting

Merk op dat we **gebied** alléén tussen de nulpunten hebben getekend. De grafiek van  $f$  zélf moet er apart bij getekend worden.  $\diamond$

Het is ook mogelijk verschillende gebieden in één `plot`-opdracht met de `filled`-optie te combineren door een lijst van expressies te plotten, en zo nodig een bijbehorende lijst van kleuren op te geven. Omdat de gebieden elkaar normaliter zullen overlappen, is de volgorde van belang. U moet daarmee maar wat experimenteren; zie daarvoor Opgave 9.15.

**Animaties.** Met het `display`-commando wordt een aantal grafieken samen in een plaatje getekend. Als we hierbij echter de optie `insequence=true` geven, dan worden deze grafieken niet tegelijk,

`insequence`

maar één voor één weergegeven. Dat kunnen we gebruiken om een bewegend plaatje te maken.

### Voorbeeldsessie

```
> f := (x,t) -> exp(-x/30)*cos(x/5-3*t);
      f := (x, t) -> e(-1/30 x) cos( $\frac{1}{5}x - 3t$ )
> plotlijst :=
  [seq( plot(f(x,t), x=0..120 ), t=0..evalf(2*Pi), 0.1 )]:
> with(plots):
> display(plotlijst, insequence=true );
Alternatief:
> animate( plot, [f(x,t), x=0..120], t=0..evalf(2*Pi), frames=50 );
(plaatjes niet getekend)
```

### Toelichting

Denk aan de dubbele punt achter het statement waarmee `plotlijst` wordt gemaakt!

`animate`

Een alternatief voor `display` met de `insequence`-optie is `animate`. In dit voorbeeld geeft dat vrijwel hetzelfde effect als het `display`-commando. Als het resultaat van `animate` wat schokkerig is, moet het aantal `frames` worden verhoogd.  $\diamond$

`animatecurve`

Raadpleeg `?plots`, `animate` voor meer interessante voorbeelden. In dit kader kunnen we ook nog het commando `animatecurve` noemen (in de bibliotheek `plots`). Probeer maar eens

```
plots[animatecurve]( [cos(t),sin(5*t),t=0..2*Pi],
                      frames=150 );
```

vergelijk de voorbeeldsessie op blz. 117.

### Opgave 9.1

De functie  $f$  met domein  $\mathbb{R} \setminus \{0\}$  is gegeven door:  $f : x \rightarrow \frac{1}{x^2} e^{\frac{1}{x}}$ .

- (a) Onderzoek  $\lim_{x \downarrow 0} f(x)$ ,  $\lim_{x \uparrow 0} f(x)$  en  $\lim_{x \uparrow 0} f'(x)$ ;
- (b) Onderzoek  $f$  verder en teken de grafiek van  $f$ . Neem voor  $x$  het interval  $[-4, 4]$ ;

- (c) Bereken voor welke  $a < 0$  geldt:  $\int_a^0 f(x) dx = f(a)$ .

*Aanwijzingen:*



- Raadpleeg `?limit` voor rechter- en linkerlimiet;
- Kies ook een andere range voor  $x$  om interessante details beter te kunnen bekijken;
- Zie (zo nodig) Module 5 (of `?solve`) voor onderdeel (c).

### Opgave 9.2

Teken de grafiek van  $f(x) = x^{\frac{2}{3}}$ ,  $x \in [-1, 1]$ .

*Aanwijzing:* Zie ook opgave 3.10.

Wat gebeurt er als u eerst de bibliotheek `RealDomain` laadt?

### Opgave 9.3

Gegeven de functie  $f(x) = x^3 - 2x$  en de lijn  $\ell$  door de punten  $(-2, -4)$  en  $(2, 4)$ .

- (a) Teken de grafiek van  $f$  en de lijn  $\ell$  in één plaatje. neem voor de  $x$ -waarden het interval  $[-3, 3]$ .
- (b) Bepaal alle  $x$ -waarden in het interval  $[-3, 3]$  waar de raaklijn aan de grafiek van  $f$  evenwijdig is aan de lijn  $\ell$ .
- (c) Voeg deze raaklijn(en) toe aan het plaatje van onderdeel (a).

### Opgave 9.4

Maak een tekening van het *gesloten* vierkant met de hoekpunten  $(-1, -1)$ ,  $(-1, 1)$ ,  $(1, -1)$ ,  $(1, 1)$ . Gebruik `textplot` om de letters **A**, **B**, **C**, **D** bij de hoekpunten te zetten.

### Opgave 9.5

Gegeven is de functie  $f : x \mapsto \frac{(x+3)^3}{3x^2}$  met domein  $\mathbb{R} \setminus \{0\}$ .

Ten opzichte van een assenstelsel  $Oxy$  is  $K$  de grafiek van  $f$ .

- (a) Voer de functie  $f$  in Maple in.
- (b) Stel een vergelijking op voor de scheve asymptoot van  $f$ .  
*Aanwijzing:* `expand( f(x) )`.
- (c) Bepaal de extrema en buigpunten van  $f$ .
- (d) Teken de grafiek van  $f$ , waarbij u voor  $x$  het interval  $[-10, 15]$  en voor  $y$  het interval  $[-2, 15]$  neemt.
- (e)  $V$  is het vlakdeel dat begrensd wordt door  $K$ , de  $x$ -as en de lijn  $x = -1$ . Bereken de oppervlakte van  $V$ . Gebruik eventueel `simplify` om een mooi antwoord te krijgen.

### Opgave 9.6

Ten opzichte van een assenstelsel  $Oxy$  is de kromme  $K$  gegeven door  $x = t^2$  en  $y = te^{t+1}$ , waarbij  $t \in \mathbb{R}$ . Stel een vergelijking op voor de asymptoot van  $K$  en teken dat deel van  $K$  dat zich bevindt in het deel van het vlak met  $x$  tussen 0 en 6, en met  $y$  tussen  $-1$  en 5. Als uw grafiek knikken vertoont, kunt u de grenzen voor  $t$  veranderen en/of `numpoints` een geschikte waarde geven (zie `?plot[options]`).

### Opgave 9.7

Gegeven is met domein  $[0, \infty)$  de functie  $f : x \mapsto 1 + x - 2\sqrt{x}$ .

- (a) Onderzoek  $f$  en teken de grafiek van  $f$  in een rechthoekig assenstelsel  $Oxy$ .
- (b) Bereken de oppervlakte van het vlakdeel ingesloten door de lijn  $y = 1$  en de grafiek van  $f$ .

Het punt  $P(p, f(p))$  met  $0 < p < 1$  ligt op de grafiek. De raaklijn in  $P$  aan de grafiek van  $f$  snijdt de  $x$ -as in  $A$  en de  $y$ -as in  $B$ .

- (c) Teken de grafiek en de raaklijn voor  $p = \frac{1}{2}$
- (d) Bewijs dat  $OA + OB = 1$  voor alle  $p \in (0, 1)$ .

### Opgave 9.8

Maak een plot van de kromme  $K$  gedefinieerd door de vergelijking

$$y^2 + \frac{1}{3}x^3 - \frac{1}{2}x^2 = a$$

voor  $a = 0$ ,  $a = -\frac{1}{12}$  en  $a = -\frac{1}{6}$ .

Merk op dat `implicitplot` voor  $a = 0$  geen fraai resultaat geeft. Maak een mooier plaatje door voor deze waarde  $y$  op te lossen uit de vergelijking en deze als functie van  $x$  te tekenen.

### Opgave 9.9

Gegeven de reeks  $\sum_{n=1}^{\infty} \frac{1}{n^2}$ .

Maak een tekening van de partiële sommen  $s_k$ ,  $k = 1, \dots, n$  als losse punten getekend. Neem  $n$  zo groot dat het plaatje suggereert dat de reeks convergeert.

### Opgave 9.10

Voor elke  $p \in [0, 4]$  is gegeven de functie

$$f_p : x \mapsto 2 \sin^2 x - p \sin x,$$


met domein  $[0, \pi]$ . Ten opzichte van een assenstelsel  $Oxy$  is  $K_p$  de grafiek van  $f_p$ . Teken in één figuur de grafieken  $K_p$  voor  $p = 0, p = 0.25, p = 0.5, \dots, p = 4$ . Neem voor  $x$  het interval  $[0, \pi]$  en voor  $y$  het interval  $[-2, 2]$ . Zet de teksten  $p = 0, p = 1, p = 2, p = 3, p = 4$  op de juiste plaatsen bij de grafieken.

*Aanwijzing:* Definieer  $f$  als een functie van twee variabelen ( $p$  en  $x$ ). Gebruik een `seq`-opdracht (zie §8.1) om een lijst van te plotten functies te maken

### Opgave 9.11

Beschouw de functie  $f : \mathbb{C} \rightarrow \mathbb{C}$  met  $f(z) = \frac{1}{z}$ .

Maak een tekening van de beelden  $f(H_a)$ , waarbij  $H_a$  de horizontale lijnen zijn, gegeven door:  $z = x + ai$ , met  $-5 \leq x \leq 5$ ,  $a$  constant (neem  $a = -5, -4, \dots, +4, +5$ ; zie figuur). Teken in hetzelfde plaatje de beelden van de verticale lijnen  $V_a$ . Kies voor  $f(H_a)$  (bv) `color=black` en voor  $f(V_a)$  `color=red`.



Teken  $f(H_a)$  en  $f(V_a)$  als geparametriseerde krommen (zie §9.1).

*Aanwijzing:* Kies `axes=BOXED` om ook lijnen te kunnen zien die (gedeeltelijk) met de normale assen samenvallen.

### Opgave 9.12

Met het commando `read("a:meting1.dat");` worden twee lijsten met ‘meetgegevens’ gemaakt (iedere keer als u dit commando geeft een andere lijst<sup>29</sup>). De  $x$ -waarden liggen in het interval  $[0, 5]$ . Let op dat de beide lijsten met *hoofdletters* (X en Y) zijn aangegeven.

- (a) Maak een tekening waarin de meetgegevens als (losse) kruisjes zijn weergegeven. Gebruik `pointplot( )` (denk er aan de `plots`-bibliotheek te laden); gebruik `[seq( )]` of `zip` om de gegevens in de juiste vorm te zetten.


<sup>29</sup>Hier wordt gedaan alsof het bestand `meting1.dat` op een diskette staat. Bedoeld is het bestand dat in Appendix 1 beschreven is.

- (b) Maak ook een tekening waarbij de meetgegevens door lijnstukken met elkaar verbonden zijn. Denk eraan dat de  $x$ -coördinaten in  $X$  niet geordend zijn.

Zie verder opgave 16.4 voor het bepalen van een gladde functie die zo goed mogelijk aansluit bij de meetpunten.

### Opgave 9.13

Probeer een parametervoorstelling te vinden voor de onderstaande kromme.



### Opgave 9.14

Bij een bepaald type chemische reactie wordt de reactiesnelheid  $v$  als functie van de concentratie  $x$  gegeven door  $v(x) = \frac{v_{\max} x}{k_M + x}$ , met  $v_{\max}$  en  $k_M$  positieve constanten.

Teken een grafiek van  $v(x)$ , waarbij u op de horizontale as een schaalverdeling met veelvouden van  $k_M$  (tot bijvoorbeeld  $10 k_M$ ) en op de verticale as  $\frac{1}{2} v_{\max}$  en  $v_{\max}$  aangeeft.

Teken ook de horizontale lijn  $v = v_{\max}$  als stippellijn.

*Aanwijzing:* Natuurlijk moet u waarden voor  $v_{\max}$  en  $k_M$  kiezen om een plaatje te kunnen tekenen. Maakt het uit wat u kiest? Zorg er in elk geval voor dat in de tekening niet zichtbaar wordt wat u hebt gekozen.

### Opgave 9.15

Kleur het gebied in dat wordt ingesloten door de grafieken van de functies

$$x^2 - 2x + 1 \quad \text{en} \quad 2x$$

*Aanwijzing:* Gebruik de optie `filled=true` Dat kan ook als u een lijst van twee expressies plot, en een lijst van twee kleuren opgeeft. Eén van de kleuren mag `white` zijn; experimenteer met de volgorde van de expressies en van de kleuren.